MathComp-Analysis

Measurable Functions and Simple Functions

Integral (only formal definitions)

Monotone Convergence Theorem

Conclusions

Formalization of the Lebesgue Integral in MathComp-Analysis

Progress Report

Reynald Affeldt (joint work with Cyril Cohen, Inria)

National Institute of Advanced Industrial Science and Technology (AIST), Tokyo, Japan

November 22, 2021

Motivation

Why formalize the Lebesgue integral?

1. Develop probability theory on top of Coq(/MathComp)
2. More generally: development of reusable machinery for analysis on top of MathComp

Approach:

- Stick to a standard presentation (a standard textbook should serve as a documentation) and engineer maintainable proofs (à la MathComp)

This presentation:

- Progress report about the formalization of the Lebesgue integral
- As an illustration: a look at the proof of the monotone convergence theorem (単調収束定理)

Outline

1. MathComp-Analysis
2. Measurable Functions and Simple Functions
3. Integral (only formal definitions)
4. Monotone Convergence Theorem
5. Conclusions

MathComp-Analysis

MathComp-Analysis adds to MathComp several mathematical structures for classical analysis [ACK+20, ACR18].



See `https://github.com/math-comp/analysis` PR# 371 and PR# 404 for this presentation.

The Lebesgue Measure in MathComp-Analysis

Our Starting Point

Formal construction of the Lebesgue measure by extension of an algebra of sets [AC21]. This includes:

- Formalization of *measurable types* whose sets form a σ-algebra (完全加法族)
  - Coq type: `measurableType`
- Formalization of *measures*
  - Coq type: `mu : {measure set T → R̄}` with a `measurableType T` and a `realType R`
- The Lebesgue measure
- Last but not the least: library lemmas
  - to deal with extended real numbers (拡大実数, standard definition)
  - to deal with sequences of reals and extended real numbers
  - to deal with infinite sums (extends bigop.v), etc.

Outline

1. MathComp-Analysis
2. Measurable Functions and Simple Functions
3. Integral (only formal definitions)
4. Monotone Convergence Theorem
5. Conclusions

Measurable Function (可測関数)

- A function with domain D is *measurable* when the preimage of any measurable set is measurable:
- ```
  Definition measurable_fun (T U : measurableType)
      (D : set T) (f : T → U) :=
    ∀ Y, measurable Y → measurable ((f @^-1 Y) ∩ D).
  ```
- There are **many** lemmas to prove about measurable functions to prove Fatou's lemma or the dominated convergence theorem (優収束定理)
  - In particular, the theory of limit superior and limit inferior (上極限と下極限)
  ```

Simple Function (単関数)

- A *simple function* f is defined by a sequence of pairwise-disjoint and measurable sets $A_0, \dots A_{n-1}$ and a sequence of elements $a_0, \dots, a_{n-1}$ such that $f(x) = \sum_{k=0}^{n-1} a_k \mathbf{1}_{A_k}(x)$ ($\mathbf{1}_{A_k}$ = 指示関数).
- Formalized as a telescope with a `uniq` range:

```
Variables (T : measurableType) (R : realType).
Record t := mk {
  f :> T → R ;
  rng : seq R ;
  uniq_rng : uniq rng ;
  full_rng : f @ setT = [set x | x ∈ rng] ;
  mpi : ∀ k, measurable (f @^-1 [set rng`_k]) }.
```

- This gives the types
  `sfun T R` of simple functions and
  `nnsfun T R` of non-negative simple functions

Illustration: Approximation Theorem

For any (1) measurable set D, any (2) function f that is (3) measurable and (4) non-negative, there exists a (5) sequence of non-negative simple functions g that is (6) non-decreasing and that (7) converges towards f.

```
Variables (D : set T) (mD : measurable D). (*1*)
Variables f : T → R̄. (*2*)
Hypothesis mf : measurable_fun D f. (*3*)
Hypothesis f0 : ∀ t, D t → 0 ≤ f t. (*4*)

Lemma approximation : ∃ g : (nnsfun T R)^ℕ, (*5*)
  nondecreasing_seq g (*6*)
  ∧ (∀ x, D x → EFin \o g^~x ⟶ f x). (*7*)
```

Approximation Theorem

Proof Idea



Figure: Approximation of function using simple functions

```
Definition approx_fun : (T → R)^ℕ := fun n x ⇒
  \sum_(k < n * 2 ^ n) k%:R * 2 ^- n * (x ∈ A n k)%:R
  + n%:R * (x ∈ B n)%:R.
```

Outline

1. MathComp-Analysis
2. Measurable Functions and Simple Functions
3. Integral (only formal definitions)
4. Monotone Convergence Theorem
5. Conclusions

Integral of a Non-negative Function

- Integral of a simple function:

```
Variables (D : set T) (f : sfun T R).
Let n := ssize f.
Let A := SFun.pi f.
Let a := SFun.rng f.
Definition sintegral : R̄ := \sum_(k < n) (a`_k)%:E * mu (A k ∩ D).
```

- Integral of a non-negative function:

$$\int_D f \mathrm{d}\mu \stackrel{def}{=} \sup_g \left\{ \int_D g \mathrm{d}\mu \;\middle|\; \begin{array}{l} g \text{ non-negative simple function} \\ \leq f \text{ over } D \end{array} \right\}$$

```
Definition nnintegral D (f : T → R̄) :=
  ereal_sup [set sintegral mu D g | g in
    [set g : nnsfun T R | ∀ x, D x → (g x)%:E ≤ f x]].
```

Integral of a Function

Integral of a (non-necessarily non-negative) function:

```
Definition integral D (f : T → R̄) :=
  nnintegral D (f ⁺) - nnintegral D (f ⁻).
```

- $\texttt{f}^{\,+} \stackrel{def}{=} \lambda x.\, \max(\texttt{f}(x), 0)$
- $\texttt{f}^{\,-} \stackrel{def}{=} \lambda x.\, \max(-\texttt{f}(x), 0)$

Outline

1. MathComp-Analysis
2. Measurable Functions and Simple Functions
3. Integral (only formal definitions)
4. Monotone Convergence Theorem
5. Conclusions

Monotone Convergence Theorem (単調収束定理)

Overview

Informal: For any non-decreasing sequence of non-negative measurable functions g_n, we have $\int_D (\lim g_n) \mathrm{d}\mu = \lim(\int_D g_n \mathrm{d}\mu)$

The proof of the monotone convergence theorem is in 3 steps:

1. Prove that it holds for simple functions (Lemma 1)
2. Prove that it holds for simple functions converging to a measurable function (Lemma 2)
3. Prove that it holds for measurable functions (Theorem)

We will only look at the formal proof of the Theorem and only state (formally) the Lemmas

Lemma 1

for the monotone convergence theorem

For any (1) measurable set D, any (2) sequence of non-negative simple functions g that is (3) nondecreasing and that (5) converges to a (4) non-negative simple function f, we have

$$\int_D f \mathrm{d}\mu = \lim_{n\to\infty} \int_D g_n \mathrm{d}\mu.$$

```
Variables (D : set T) (mD : measurable D). (*1*)
Variable g : (nnsfun T R)^ℕ. (*2*)
Hypothesis nd_g : ∀ x, D x → nondecreasing_seq (g ^~ x).
Variable f : nnsfun T R. (*4*)
Hypothesis gf : ∀ x, D x → g ^~ x ⟶ f x. (*5*)

Lemma nd_sintegral_lim :
  sintegral mu D f = lim (sintegral mu D \o g).
```

Lemma 2

for the monotone convergence theorem

For any (1) measurable set D, any (2) function f that is (3) non-negative and (4) measurable, any (5) sequence of non-negative simple functions g that is (6) non-decreasing and (7) converging towards f, we have

$$\int_D f \mathrm{d}\mu = \lim_{n \to \infty} \int_D g_n \mathrm{d}\mu.$$

```
Variables (D : set T) (mD : measurable D). (*1*)
Variable f : T → R̄. (*2*)
Hypothesis f0 : ∀ x, D x → 0 ≤ f x. (*3*)
Hypothesis mf : measurable_fun D f. (*4*)
Variable g : (nnsfun T R)^ℕ. (*5*)
Hypothesis nd_g : ∀ x, D x → nondecreasing_seq (g ^~ x). (*6*)
Hypothesis gf : ∀ x, D x → EFin \o g ^~ x ⟶ f x. (*7*)

Lemma nd_ge0_integral_lim :
  integral mu D f = lim (sintegral mu D \o g).
```

Monotone Convergence Theorem (単調収束定理)

For (1) any measurable set D and any (3) non-decreasing sequence of functions (2) $g_n : T \to \overline{\mathbb{R}}$ that are (4) measurable and (5) non-negative, we have

$$\int_D \left(\lim_{n\to\infty} g_n\right) \mathrm{d}\mu = \lim_{n\to\infty} \int_D g_n \mathrm{d}\mu.$$

```
Variables (D : set T) (mD : measurable D). (*1*)
Variable g : (T → R̄)^ℕ. (*2*)
Hypothesis nd_g : ∀ x, D x → nondecreasing_seq (g ^~ x). (*3*)
Hypothesis mg : ∀ n, measurable_fun D (g n). (*4*)
Hypothesis g0 : ∀ n x, D x → 0 ≤ g n x. (*5*)

Lemma monotone_convergence :
  integral mu D (fun x ⇒ lim (g ^~ x)) =
    lim (fun n ⇒ integral mu D (g n)).
```

Monotone Convergence Theorem

Easy direction

$$\lim_{n\to\infty}\int_D g_n \mathrm{d}\mu \leq \int_D \left(\lim_{n\to\infty} g_n\right) \mathrm{d}\mu$$

```
lim (fun n ⇒ integral mu D (g n)) ≤ integral mu D (fun x ⇒ lim (g ^~ x))
```

The proof is by appealing to properties of sequence of extended real numbers and to the fact that the integral is monotone:

```
(* for measurable, non-negative functions *)
Lemma ge0_le_integral : (∀ x, D x → f1 x ≤ f2 x) →
  integral mu D f1 ≤ integral mu D f2.
```

Indeed, we can use `ge0_le_integral` to show that the sequence on the LHS is non-decreasing and to show that each term is bounded by the RHS.

Monotone Convergence Theorem

Easy direction in Coq

```
Lemma monotone_convergence :
 integral mu D f = lim (fun n => integral mu D (g n)).
Proof.
apply/eqP; rewrite eq_le; apply/andP; split; last first.
 have nd_int_g : nondecreasing_seq (fun n => integral mu D (g n)).
  move=> m n mn; apply: ge0_le_integral => //.
   by move=> *; exact: g0.
   by move=> *; exact: g0.
   by move=> *; exact: nd_g.
 have ub n : integral mu D (g n) <= integral mu D f.
  apply: ge0_le_integral => //.
  - by move=> *; exact: g0.
  - move=> x Dx; apply: ereal_lim_ge => //; first exact/is_cvg_g.
   by apply: nearW => k; apply/g0.
  - move=> x Dx; apply: ereal_lim_ge => //; first exact/is_cvg_g.
   near=> m.
   have nm : (n <= m)%N by near: m; exists n.
   exact/nd_g.
 by apply: ereal_lim_le => //; [exact: ereal_nondecreasing_is_cvg|exact: nearW].
...
```

Monotone Convergence Theorem

Difficult direction (1/2)

$$\int_D \underbrace{\left(\lim_{n\to\infty} g_n\right)}_{f} \mathrm{d}\mu \leq \lim_{n\to\infty} \int_D g_n \mathrm{d}\mu$$

```
integral mu D (fun x ⇒ lim (g ^~ x)) ≤ lim (fun n ⇒ integral mu D (g n))
```

The idea is to build a sequence of non-negative simple functions h_n (next slide) that is non-decreasing and such that $h_n \leq g_n$ and $\lim_{n\to\infty} h_n = f$.

Then we can use Lemma 2 to show

$$\int_D f \mathrm{d}\mu = \lim_{n\to\infty} \int_D h_n \mathrm{d}\mu$$

which leads to

$$\lim_{n\to\infty} \int_D h_n \mathrm{d}\mu \leq \lim_{n\to\infty} \int_D g_n \mathrm{d}\mu.$$

Monotone Convergence Theorem

Difficult direction (1/2 in Coq)

```
Lemma monotone_convergence :
  integral mu D f = lim (fun n => integral mu D (g n)).
Proof.
apply/eqP; rewrite eq_le; apply/andP; split; last first.

  ... (easy direction) ...

rewrite (@nd_ge0_integral_lim _ point _ mu _ _ _ _ max_g2) //; last 3 first.
  - move=> t Dt; apply: ereal_lim_ge => //; first exact/is_cvg_g.
    by apply: nearW => n; apply: g0.
  - by move=> t Dt m n mn; apply/lefP/nd_max_g2.
  - by move=> x Dx; exact: cvg_max_g2_f.
apply: lee_lim.
- apply: is_cvg_sintegral => //.
  by move=> t Dt m n mn; exact/lefP/nd_max_g2.
- apply: ereal_nondecreasing_is_cvg => // n m nm; apply: ge0_le_integral => //.
  + by move=> *; apply: g0.
  + by move=> *; apply: g0.
  + by move=> *; apply/nd_g.
- apply: nearW => n.
  rewrite ge0_integralE//; last by move=> *; apply: g0.
  by apply: ereal_sup_ub; exists (max_g2 n) => // t; exact: max_g2_g.
Grab Existential Variables. all: end_near. Qed.
```

Monotone Convergence Theorem

Difficult Direction (2/2)

Reminder: we want simple functions h_n s.t. $\lim_{n\to\infty} h_n = f$

We approximate (in the sense of the approximation Theorem) each measurable function g by a function g_2 and create a sequence of functions `h`

```
Local Definition g2 n : (T → R)^ℕ :=
  approx_fun D (g n).
Local Definition h : (T → R)^ℕ :=
  fun n t ⇒ \big[maxr/0]_(i < n) (g2 i n) t.
```

- h_n non-decreasing? Yes, essentially because each `g2` is
- $h_n \leq g_n$? Yes, essentially because $\mathtt{g2}_n \leq g_n$
- $\lim_{n\to\infty} h_n = f$? ...

Monotone Convergence Theorem

Difficult Direction (2/2)

... $\lim_{n\to\infty} h_n = \lim_{n\to\infty} g_n$?

- $\lim_{n\to\infty} h_n \leq \lim_{n\to\infty} g_n$ is easy
- $\lim_{n\to\infty} g_n \leq \lim_{n\to\infty} h_n$
  - Suppose that the RHS is $< +\infty$
  - It suffices to prove:

    `\∀ n \near \oo, g n t ≤ lim (EFin \o h ^~ t)`

    - `g n t` is $+\infty$:
      then `(approx_fun D (g n))^~ t` diverges,
      then `lim (EFin \o g2 n ^~ t) = +oo`,
      then `lim (EFin \o h ^~ t) = +oo`
    - `g n t` $< +\infty$:
      then `(approx_fun D (g n))^~ t` converges to
      `g n t`,
      then `lim (EFin \o g2 n ^~ t) = g n t`,
      we conclude because each `g2` is smaller or equal to `h`

Monotone Convergence Theorem

The Last Part of Reasoning in Coq

```
Local Lemma cvg_max_g2_f t : D t → EFin \o max_g2 ^~ t ⟶ f t.
Proof.
move⇒ Dt; have /cvg_ex[l g_l] := @is_cvg_max_g2 t.
suff : l == f t by move⇒ /eqP ←.
rewrite eq_le; apply/andP; split.
  by rewrite /f (le_trans _ (lim_max_g2_f Dt)) // (cvg_lim _ g_l).
have := lee_pinfty l; rewrite le_eqVlt ⇒ /predU1P[→|loo].
  by rewrite lee_pinfty.
rewrite −(cvg_lim _ g_l) //= ereal_lim_le ⇒ //; first exact/is_cvg_g.
near⇒ n.
have := lee_pinfty (g n t); rewrite le_eqVlt ⇒ /predU1P[|] fntoo.
− have h := dvg_approx_fun Dt fntoo.
  have g2oo : lim (EFin \o g2 n ^~ t) = +oo%E.
    apply/cvg_lim ⇒ //; apply/dvg_ereal_cvg.
    under [X in X ⟶ _]eq_fun do rewrite nnsfun_approxE.
    exact/(nondecreasing_dvg_lt _ h)/lef_at/nd_approx_fun.
  have → : lim (EFin \o max_g2 ^~ t) = +oo%E.
    by have := lim_g2_max_g2 t n; rewrite g2oo lee_pinfty_eq ⇒ /eqP.
  by rewrite lee_pinfty.
− have approx_fun_g_g := cvg_approx_fun (g0 n) Dt fntoo.
  have ← : lim (EFin \o g2 n ^~ t) = g n t.
    have /cvg_lim ← // : EFin \o (approx_fun D (g n))^~ t ⟶ g n t.
      move/(@cvg_comp _ _ _ _ EFin) : approx_fun_g_g; apply.
      by rewrite −(@fineK _ (g n t))// ge0_fin_numE// g0.
    rewrite (_ : _ \o _ = EFin \o (approx_fun D (g n))^~ t)// funeqE ⇒ m.
    by rewrite [in RHS]/= −(nnsfun_approxE point).
  exact: (le_trans _ (lim_g2_max_g2 t n)).
Grab Existential Variables. all: end_near. Qed.
```

Outline

1. MathComp-Analysis
2. Measurable Functions and Simple Functions
3. Integral (only formal definitions)
4. Monotone Convergence Theorem
5. Conclusions

Related Work

- Formalization of the Lebesgue integral up to Fatou's lemma in Coq on top of Coquelicot [BCF+21]
  - Main differences: no Lebesgue measure, addition of extended real numbers not associative
- In HOL4 [MHT10]
- Recent formalization of the Lebesgue measure in Mizar [End20]
- Rich formalization of integration theory in Lean [vD21]
- Also in Isabelle/HOL [HH11]

Conclusion

- We have developed the Lebesgue measure and integral
  - in Coq (one may claim this is the first such framework)
  - up to the dominated convergence theorem (優収束定理)
  - the salient different with other proof assistants is likely to be the construction of the Lebesgue measure (not this talk)
- We have been doing so by sticking to standard definitions, standard constructions, and a standard textbook
- Recent work: product measure, Fubini's theorem (wip)
- Future work:
  - Probability theory (i.e., extend InfoTheo [AGS20] to the continuous case)
  - Application to probabilistic programming (i.e., extend Monae [AGNS21])

Reynald Affeldt and Cyril Cohen, *Formalization of the Lebesgue measure in MathComp-Analysis*, The Coq Workshop 2021, online, July 2, 2021, Jul 2021, p. 2 pages.

Reynald Affeldt, Cyril Cohen, Marie Kerjean, Assia Mahboubi, Damien Rouhling, and Kazuhiko Sakaguchi, *Competing inheritance paths in dependent type theory: a case study in functional analysis*, IJCAR 2020, LNAI, vol. 12167(2), Springer, Jul 2020, pp. 3–20.

Reynald Affeldt, Cyril Cohen, and Damien Rouhling, *Formalization techniques for asymptotic reasoning in classical analysis*, Journal of Formalized Reasoning **11** (2018), no. 1, 43–76.

Reynald Affeldt, Jacques Garrigue, David Nowak, and Takafumi Saikawa, *A trustful monad for axiomatic reasoning with probability and nondeterminism*, Journal of Functional Programming **31** (2021), no. E17.

Reynald Affeldt, Jacques Garrigue, and Takafumi Saikawa, *A library for formalization of linear error-correcting codes*, Journal of Automated Reasoning **64** (2020), 1123–1164.

Sylvie Boldo, François Clément, Florian Faissole, Vincent Martin, and Micaela Mayero, *A Coq formalization of Lebesgue Integration of nonnegative functions*, Tech. Report RR-9401, Inria, 2021.

Noboru Endou, *Reconstruction of the one-dimensional Lebesgue measure*, Tech. report, National Institute of Technology, Gifu College, 2020, Formalized Mathematics 28(1):93–104.

Johannes Hölzl and Armin Heller, *Three chapters of measure theory in Isabelle/HOL*, ITP 2011, LNCS, vol. 6898, Springer, 2011, pp. 135–151.

Tarek Mhamdi, Osman Hasan, and Sofiène Tahar, *On the formalization of the lebesgue integration theory in HOL*, First International Conference on Interactive Theorem Proving (ITP 2010), Edinburgh, UK, July 11–14, 2010, Lecture Notes in Computer Science, vol. 6172, Springer, 2010, pp. 387–402.

Floris van Doorn, *Formalized Haar measure*, 12th International Conference on Interactive Theorem Proving (ITP 2021) June 29–July 1, 2021, Rome, Italy (Virtual Conference), LIPIcs, vol. 193, Schloss Dagstuhl - Leibniz-Zentrum für Informatik, 2021, pp. 18:1–18:17.